Fermilab

TM-1256
1733.080
1760.000

CONSECUTIVE QUENCHES AND THE SAFETY LEADS

M. Kuchnir

April 30, 1984

Consecutive Quenches and the Safety Leads

M. Kuchnir

April 30, 1984

The safety leads of the Energy Saver were designed to handle occasional quenches. In order to save capital investment in another system of plumbing they involve no gas cooling. In their design optimisation was sought on the ratio of load capability (for a single quench) to steady state heat leak into the liquid helium environment. Reference 1, describes the design considerations and tests on a prototype made out of constanta. The actual production models are made out of 304 Stainless Steel rods 13/16" diameter by 48.66" long[2] wrapped in Kapton film for electrical insulation.

Here the behaviour of production models relevant for consecutive quenches is analysed. The quenches are characterized by current pulses of the shape

$$I(t) = I_o\, e^{-t/12.}$$

Since the time constant of 12. seconds is much smaller than the cooling time for recovery, the adiabatic approximation for the temperature increase in an element of length dx of the lead during a quench is accurately given by the energy balance equation:

$$\mu\ s\ c(T)\ dT = \rho(T)\ s^{-1}\ I(t)^2\ dt$$

where:

μ = 7.86 g/cm^3 is the density of stainless steel[3]

$s = \frac{\pi}{4}\,(13/16)^2\ \text{in}^2 = 3.345\ \text{cm}^2$ is the cross section

$c(T)$ is the specific heat of stainless steel[4] $(J/_{g.K})$

$\rho(T)$ is the electrical resistivity of stainless steel[5] (Ω.cm)

T is the temperature of the element (K)

t is time (s)

rearranging and integrating:

$$\mu\ s^2 \int_{T_{before}}^{T_{after}} c(T)/\rho(T)\ dT = \int_0^{\infty} I_o^{\,2}\ e^{-2t/12.} dt = 6.I_o^{\,2}$$

The right hand side of the equations is usually refered to as the number of Miits (10^6 A^2.s or J/μΩ) forced into the lead. Based on quench data, P. Martin[6] has suggested a reduction of the value 6 I_o^2 to 5.6 I_o^2 in order to account for the initial deposition of energy in the magnet.

Figure 1 is a calculation of the integral

$$\mu\ s^2 \int_0^T c(T)/\rho(T)\ dT \qquad J/\mu\Omega$$

and it allow us to find out the temperature excursion of an element of length given its initial temperature and the intensity of the quench either in J/μΩ (Miits) or the initial current using the scale on the right margin. This scale has not been corrected for the 5.6/6.0 factor in the number of Miits. The same Figure 1 allows us to evaluate the maximum current allowable if the temperature of the hottest spot in the lead is known and the maximum safe value (600. K for the integrity of the Kapton film) is stipulated. Since the electrical resistance of the lead can be estimated from quench data[6], it is useful to have a calculated limit for the temperature of the hottest spot in the lead as a function of the electrical resistance. This is given in Figure 2. The calculation leading to this graph is explained further on.

The temperature profile of the lead is a very slow function of time between quenches as well as during warm up and cool down of the Energy Saver, since it is cooled only by conduction to its extremities. The superinsulation reduces infrared radiation to a negligible effect. We therefore neglect this effect.

In order to calculate the change of the temperature profile with time, we set up the following finite element difference equation for the energy flow:

$$Q(n,n-1) - Q(n+1,n) = h(n)\cdot(\ T_n(t+dt)-T_n(t)\ )\ /\ dt$$

where:

$Q(n,n-1)$ = heat conducted from element n to element n-1
$Q(n+1,n)$ = heat conducted from element n+1 into element n
$h(n) = \mu\ s\ dx\ c(T_n)$ = change in heat capacity of element n during dt

Let $\kappa(T_{n,n-1})$ be the thermal conductivity of stainless steel[7] (w/cm.K) at the temperature $T_{n,n-1} = (T_n+T_{n-1})/2$ between elements n and n-1, then

$$Q(n,n-1) = s\cdot\kappa(T_{n,n-1})\cdot(T_n - T_{n-1})\ /\ dx$$

$$Q(n+1,n) = s\cdot\kappa(T_{n+1,n})\cdot(T_{n+1} - T_n)\ /\ dx$$

For boundary conditions we impose T_1=4.6 K and T_{62}=298. K, the length of the lead being divided in 62 elements, each 2.0 cm long (dx=2. cm). The initial condition T_n=300. K for n=2, 3,..., 61 at t=0 would be typical for the cooldown problem. The above expressions permit the calculation of the temperature profile at time t=dt, and by repetition at any multiple of dt which was chosen to be dt=6.0 s. Figure 3 shows this condition as it evolves into the steady state profile. It is interesting to note that although thermal difusivity is the relevant property in this heat propagation problem, its use leads to an erroneous linear temperature profile for the steady state condition. One has to handle separately, as we did above, the thermal conductivity and the specific heat for a proper simulation.

A public fortran program, QCHWAIT/UN=92532, in the Cyber was written to carry out the simulation. It uses approximately 1 minute of cpu time to simulate 10 hours of lead behaviour. This program is presented in Appendix 1. The output of this program for the quenches observed on April 9-10, 1984 is plotted in Figure 4. The development of the steady state temperature profile for a safety lead initially at room temperature presented in Figure 3 was also obtained with this program. The printing is done on intervals of approximately 15 minutes, it includes the time (decimal hours), the highest temperature (K) in the profile, the distance from the cold end (cm) at which it occurs and a series of temperatures (K) corresponding to positions separated by 10 cm. In the process of developing this program, we compiled in another program QCHWTST/UN=92532 the following properties of stainless steel: thermal difusivity, specific heat, electrical resistivity, thermal conductivity, quench capability and inverse quench capability.

By using the electrical resistivity and the temperature profile at print out time, QCHWAIT or its later version calculates the total resistance

$$R = \sum_{n=1}^{62} \rho(T_n)\ s^{-1}\ dx$$

Plotting this total resistance as a function of the hottest point temperature for the three quenches of April 9, 1984 and several other quenches (2.0 kA, 3.0 kA, 3.5 kA and 7.7 kA) starting from steady state condition we obtain Figure 2. An envelope curve can be drawn in it. This curve is the temperature of the hottest spot possible for the given resistance. If the resistance of the safety lead can be measured in place, remotely, with a phase-locked-ac-milliohmeter for instance, or calculated from quench data this curve might be very useful as mentioned above in the first reference to Figure 2.

A plot of the temperature of the hottest spot as a function of time is presented in Figure 5. It includes the three quenches of Figure 4 starting with a delay of 4 hours and several other quenches starting from the steady state condition. It provides us with examples for estimating cooling or waiting times.

Due mostly to time limitation, this work is presented leaving a few rough edges. The following list is written while things that could be improved, expanded and polished are fresh in mind. It is a "to do" list in no particular order, if upgrading this work becomes a justifiable effort.

1. Figure 1 was calculated by hand not using the stainless steel

data functions that now are in the program, it might deviate a little from the values that these might generate for it.

2. Modify QCHWAIT for convenience and user friendliness to the accelerator operators, as well as, presenting the results in plots instead of tables.

3. Verify the performance of QCHWAIT with different values of dx and dt. The values chosen satisfy the thermodynamic principles condition

$$dx^2/2dt < \kappa/\mu c$$

at all temperatures. The calculations show good convergence and reasonable answers passing several tests. But, it is probably not the optimum compromise between simulation speed and detail.

4. Measure the electrical resistance of several safety leads and compare them with the calculated values shown in Figure 2.

5. Instrument a safety lead with thermometers and experimentally verify the results of this work.

6. Simulate the "bleeding gas fix" that was implemented in the few critical spool pieces for speedier cooling off.

REFERENCES

1. M.Kuchnir and T.H.Nicol - Safety Leads - Advances in Cryogenic Engineering 25 , 294-299, (1980).

2. Fermilab drawing No. 1620-MD-124599

3. Materials at Low Temperatures - Richard P. Reed and Alan F. Clark editors, American Society for Metals, Metals Park, Ohio 44073 (1983) p.377

4. Ibid p. 66

5. Ibid p. 164

6. Philip S. Martin, Private communication

7. Reference 3, p. 135



Figure 1. Load Capacity (Miits) for a Safety Lead



Figure 2. Hottest Spot Temperature vs. Lead Resistance



Figure 3. Temperature Profiles During Cooldown



Figure 4. Temperature Profiles for April 9, 1984 Quenches



Figure 5. Cooling of Hottest Spots for Several Quenches

Appendix 1

Simulation Program

```
      PROGRAM QCHWAIT (INPUT,OUTPUT,HISTO,DET,TAPE1=HISTO,TAPE2=DET)
C                       THIS PROGRAM SIMMULATES THE TIME DEPENDENCE OF
C     THE TEMPERATURE PROFILE OF AN ENERGY-SAVER SAFETY LEAD.840413
C
      DIMENSION T(62,150),TSS(62),QTIME(5),QCURR(5)
      DX=2.
      DT=6.
      S=3.345
      R=7.86
      DD=DT/(DX*DX*R)
C     SAMPLE HISTORY:
      ACTIME=4.0
      DATA QTIME/ 4.38, 9.45, 13.1, 34.6, 110./
      DATA QCURR/ 2720.,3552.,3552.,2800.,-1./
      NQ=1
C     QUENCH OF CONCERN:
      AIMAX=4000.
      QMAX=5.6E-6*AIMAX*AIMAX
      QM=FUNC1(300.)+QMAX
      TMAX=FUNC2(QM)
C     STEADY STATE PROFILE:
      DATA TSS/ 4.6,33.,43.,52.,60.,67.,74.,80.,86.,91.,
     1 97.,102.,107.,112.,117.,122.,126.,131.,136.,140.,
     2145.,149.,153.,158.,162.,166.,170.,174.,179.,182.,
     3186.,190.,194.,198.,202.,206.,209.,213.,216.,220.,
     4224.,227.,231.,234.,238.,242.,245.,249.,252.,256.,
     5259.,263.,266.,270.,273.,277.,280.,284.,287.,291.,
     6294.,298./
      WRITE(1,51)
      PRINT 51
   51 FORMAT(" TIME HR   TMAX     LMAX     R MOHMS T 12     T 22    T 32
     1 T 42     T 52     T 62     T 72     T 82     T 92     T102  T112  T
     2122")
C     SET INITIAL PROFILE TO STEADY STATE
      DO 1 I=1,62
      T(I,1)=TSS(I)
    1 CONTINUE
C     HISTORY HANDLER
   71 IF(QTIME(NQ)-.25-ACTIME) 10,10,11
   11 N=150
      IFQ=0
      GOTO 2
   10 N=3600.*(QTIME(NQ)-ACTIME)/DT
      IFQ=1
      IF(N.GT.150) PRINT 12
   12 FORMAT( "N>150")
      IF(QCURR(NQ).LE.0) STOP
      GOTO 2
C
C     FINITE ELEMENT ITERATION
```

```
    2 DO 3 J=2,N
      T(1,J)=4.6
      T(62,J)=298.
      DO 4 I=2,61
      T1=(T(I-1,J-1)+T(I,J-1))/2.
      DT1=T(I-1,J-1)-T(I,J-1)
      T2=(T(I+1,J-1)+T(I,J-1))/2.
      DT2=T(I+1,J-1)-T(I,J-1)
C     WRITE(2,54) I,J,T1,DT1,T2,DT2
C 54  FORMAT (2I4,4F10.1)
      C2=DD*COND(T2)/SPEC(T(I,J-1))
      C1=DD*COND(T1)/SPEC(T(I,J-1))
      T(I,J)=T(I,J-1) + C1 * DT1 + C2 * DT2
    4 CONTINUE
      TD=J-1
      L=J
    3 CONTINUE
      ACTIME=ACTIME+(TD*DT/3600.)
      TMAX=0
      R=0
      DO 5 I=1,62
      T(I,1)=T(I,L)
      R=R+DX*RESIS(T(I,L))/S
      H=I
      IF(TMAX.LE.T(I,L)) TMAX=T(I,L)
      IF(TMAX.LE.T(I,L)) HMAX=H*DX
    5 CONTINUE
      WRITE(1,50) ACTIME,TMAX,HMAX,R,T( 6,1),T(11,1),T(16,1),
     1T(21,1),T(26,1),T(31,1),T(36,1),T(41,1),T(46,1),T(51,1),
     2T(56,1),T(61,1)
   50 FORMAT(F8.2,2F8.0,F8.3,12F8.0)
      PRINT 52, ACTIME,TMAX,HMAX,R,T( 6,1),T(11,1),T(16,1)
      IF (IFQ.EQ.0) GOTO 71
C
C     EXECUTE QUENCH:
      QI=5.6E-6*QCURR(NQ)*QCURR(NQ)
      WRITE(1,53) QTIME(NQ),QCURR(NQ),QI
   53 FORMAT(F8.2,"  QUENCH WITH IO=",F6.0,"A  ===>",F6.1," MIITS")
      TMAX=0
      R=0
      DO 7 K=1,62
      QE=FUNC1(T(K,1))+QI
      T(K,1)=FUNC2(QE)
      R=R+DX*RESIS(T(K,1))/S
      H=K
      IF(TMAX.LE.T(K,1)) TMAX=T(K,1)
      IF(TMAX.LE.T(K,1)) HMAX=H*DX
    7 CONTINUE
      WRITE(1,50) ACTIME,TMAX,HMAX,R,T( 6,1),T(11,1),T(16,1),
     1T(21,1),T(26,1),T(31,1),T(36,1),T(41,1),T(46,1),T(51,1),
     2T(56,1),T(61,1)
      PRINT 52, ACTIME,TMAX,HMAX,R,T( 6,1),T(11,1),T(16,1)
   52 FORMAT(F8.2,2F8.0,F8.3,3F8.0)
      NQ=NQ+1
```

```
C     EXAMINE MAXIMUM TEMPERATURE
      QE=FUNC1(TMAX)+QMAX
      TFUT=FUNC2(QE)
      IF(TFUT.GE.650) WRITE(1,8) AIMAX,TFUT,HMAX
    8 FORMAT(8X,"A ",F5.0,"A  QUENCH WILL GENERATE ",F4.0,"K AT ",
     1F5.0,"CM")
C     EXAMINE MAXIMUM ALLOWABLE QUENCH
      QA=FUNC1(650.)-FUNC1(TMAX)
      IF(QA) 60,60,62
   60 WRITE(1,64)
   64 FORMAT(7X," EXCEEDED MAXIMUM ALLOWABLE QUENCH ")
      GOTO 71
   62 CUR2=QA/5.6E-6
      CUR=SQRT(CUR2)
      WRITE(1,66) CUR
   66 FORMAT(7X," MAXIMUM ALLOWABLE QUENCH CURRENT NOW:",F6.0,"A .")
      GOTO 71
      END
C
C
      FUNCTION FUNC1(T)
C              QUENCH CAPACITY AS FUNCTION OF TEMPERATURE    MIITS
      IF(T.LE.400.) GOTO 31
      FUNC1=-22.5556+.475556*T
      GOTO 37
   31 IF(T.LE.275.) GOTO 32
      FUNC1=-42.2222+.522222*T
      GOTO 37
   32 IF(T.LE.200.) GOTO 33
      FUNC1=-52.5+.56*T
      GOTO 37
   33 IF(T.LE.96.) GOTO 34
      FUNC1=.736963E-3*T**2.13255
      GOTO 37
   34 IF(T.LE.4.6) GOTO 35
      FUNC1=1.16459E-6*T**3.54396
      GOTO 37
   35 PRINT 36
   36 FORMAT("T<4.6 FUNC1 TRAP")
      FUNC1=0.
   37 RETURN
      END
C
      FUNCTION FUNC2(X)
C              TEMPERATURE INCREMENT DUE TO QUENCH LOAD      K
      IF(X.LE.168.) GOTO 41
      FUNC2=(X+22.5556)/.475556
      GOTO 47
   41 IF(X.LE.102.) GOTO 42
      FUNC2=(X+42.2222)/.522222
      GOTO 47
   42 IF(X.LE.60.) GOTO 43
      FUNC2=(X+52.5)/.56
      GOTO 47
```

```
  43  IF(X.LE.10.) GOTO 44
      FUNC2=(X/.736963E-3)**.468922
      GOTO 47
  44  IF(X.LE..00026) GOTO 45
      FUNC2=(X/1.16459E-6)**.282170
      GOTO 47
  45  PRINT 46
  46  FORMAT("X<.00026  -FUNC2 TRAP")
      FUNC2=0.
  47  RETURN
      END
C
      FUNCTION SPEC(T)
C                SPECIFIC HEAT OF STAINLESS STEEL  J/GK
      IF(T.LE.600.) GOTO 81
      SPEC = .49
      GOTO 89
  81  IF(T.LE.300.) GOTO 82
      SPEC = .22081519 * T ** .12285675
      GOTO 89
  82  IF(T.LE.200.) GOTO 83
      SPEC = .085829531 * T ** .29048871
      GOTO 89
  83  IF(T.LE.100.) GOTO 84
      SPEC = 4.1440952E-3 * T ** .86249648
      GOTO 89
  84  IF(T.LE.63.) GOTO 85
      SPEC = 85.005947E-6 * T ** 1.7064867
      GOTO 89
  85  IF(T.LE.35.) GOTO 86
      SPEC = 1.6692594E-6 * T ** 2.6551262
      GOTO 89
  86  IF(T.LE.4.2) GOTO 87
      SPEC = 478.40362E-6 * T ** 1.06370
      GOTO 89
  87  PRINT 88
  88  FORMAT (" T<4.2 ==> SPEC = 1.E-6")
      SPEC = 1.E-6
  89  RETURN
      END
C
      FUNCTION COND(T)
C                THERMAL CONDUCTIVITY OF STAINLESS STEEL  W/CM.K
      IF(T.LE.300.) GOTO 92
      COND = 44.707195E-3 * T ** .21222774
      GOTO 99
  92  IF(T.LE.200.) GOTO 93
      COND = .020037364 * T ** .3529301
      GOTO 99
  93  IF(T.LE.100.) GOTO 94
      COND = 11.822224E-3 * T ** .4525122
      GOTO 99
  94  IF(T.LE.60.) GOTO 95
      COND = 4.6622033E-3 * T ** .6545662
```

```
      GOTO 99
   95 IF(T.LE.40.) GOTO 96
      COND = 1.6319045E-3 * T ** .9109541
      GOTO 99
   96 IF(T.LE.20.) GOTO 97
      COND = 435.29606E-6 * T ** 1.2691866
      GOTO 99
   97 IF(T.LE.4.2) GOTO 98
      COND = 394.94084E-6 * T ** 1.301663
      GOTO 99
   98 PRINT 100
  100 FORMAT (" T<4.2 --> COND=1.E-4")
      COND=1.E-4
   99 RETURN
      END
C
      FUNCTION RESIS(T)
C              RESISTIVITY OF STAINLESS STEEL     1.E-3*OHMS*CM
      IF(T.LE.148.) GOTO 75
      RESIS = .020581593 * T ** .21410886
      GOTO 79
   75 IF(T.LE.80.) GOTO 76
      RESIS = .020916588 * T ** .21087798
      GOTO 79
   76 IF(T.LE.40.) GOTO 77
      RESIS = .035922037 * T ** .087462841
      GOTO 79
   77 RESIS=.0496
      IF (T.LT.4.2) PRINT 78
   78 FORMAT ("T<4.2 IN RESIS")
   79 RETURN
      END
      END
```